
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc160813210]Corrections required
1. [bookmark: _Toc160813211][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665]Correction to function f_TC_7_1_NR5GC_IMS1
	Function name
	f_TC_7_1_NR5GC_IMS1

	Reason for change
	First step inside current postamble will be to release the call but this will not work when UE behaviour is not to retry the call after 503 (Service Unavailable) is sent.  

	Summary of change
	Added an alternative postamble that will be chosen in case UE does not retry IMS call after 503 (Service Unavailable) is sent by SS.

	TTCN module
	IMS\NR5GC_7\IMS_NR5GC_CallControlTestcases.ttcn

	MCC160 Comment
	Accepted in principle:
The issue shall be solved at the IPCAN side (see below): f_NR5GC_IPCAN_StartProcedure needs to return the return value of f_NR5GC_IPCAN_MO_SpeechCall, f_NR5GC_IPCAN_MT_SpeechCall, f_NR5GC_IPCAN_MO_VideoCall, f_NR5GC_IPCAN_MT_VideoCall, f_NR5GC_IPCAN_AddVideo and f_NR5GC_IPCAN_ReleaseVideo so that in case of ABORT f_NR5GC_IPCAN_ProcedureEnd does not try to release a QoS flow.
In addition, as f_IMS_PTC_ImsInfo_GetContactUrlUE raises an error when the dialog is not started yet (instead of returning omit) another change in f_IMS_CC_PostambleWithCallReleaseMT is needed.

	
	


[bookmark: _Toc54888064]
Before Change:
	<<SKIPPED CODE>>
  function f_TC_7_1_NR5GC_IMS1() runs on IMS_PTC
  { // MTSI MO Voice Call / 503 Service Unavailable / 5GS
    var template (present) SipUrl v_CalleeUri := f_IMS_INVITE_CalleeUriRX();    // as used per default in the INVITE and in this test case for the ACK as the UE has not gotten any callee's contact URI
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var INVITE_Request v_InviteRequest;
    var template (value) MessageHeader v_MessageHeader503;
    var template (present) MessageHeader v_MessageHeaderACK;
    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;
    var integer v_RetryAfter := 20;   // @sic R5-210882: 60s -> 20s sic@
    var float v_RetryAfterMin := f_IMS_SetTimerToleranceMin(int2float(v_RetryAfter));
    var boolean v_InviteWithPreconditions;
    
    timer t_Timer;

    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED);
    
    f_IMS_TestBody_Set(true);

    // @siclog "Step 1" siclog@
    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall);

    // @siclog "Step 2 - 7" siclog@
    v_InviteWithPreconditions := fl_IMS_MOCallSetup_CheckInviteWithPreconditions();

    if (v_InviteWithPreconditions) {  // @sic R5-213521 sic@
      // @siclog "Step 8a" siclog@              INVITE
      v_InviteRequestWithSdp := f_IMS_MOCallSetup_NR5GC_A_4_1a_Step1(v_CalleeUri);               // @sic R5-225925, R5-236507: A_4_1 -> A_4_1a sic@
    } else {
      // @siclog "Step 8b" siclog@              INVITE
      v_InviteRequestWithSdp := f_IMS_MOCallSetup_NR5GC_A_4_2a_Step1(v_CalleeUri);               // @sic R5-236507: A_4_2 -> A_4_2a sic@
    }
    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    // @siclog "Step 9" siclog@                 503 Service Unavailable
    v_MessageHeader503 := f_IMS_CommonResponse_503_MessageHeaderTX(v_InviteRequest.msgHeader, cs_RetryAfter(v_RetryAfter));
    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_InviteRequestWithSdp.RoutingInfo);
    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine503, v_MessageHeader503)));
    
    // @siclog "Step 10" siclog@                ACK
    v_MessageHeaderACK := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine503);
    f_IMS_ReceiveRequest(car_IMS_Ack_Request(cr_ACK_Request(v_CalleeUri, v_MessageHeaderACK)));
    
    f_IMS_PTC_ImsInfo_DialogRelease();

    // @siclog "Step 11 - 13" siclog@           Check: Does the UE (not) transmit INVITE request message
    f_Delay(v_RetryAfterMin);
    
    t_Timer.start(30.0);
    alt {                     // @sic R5-216210 sic@
      [] IMS_Server.check(receive(car_IMS_Invite_Request)) {
        t_Timer.stop;
        if (v_InviteWithPreconditions) {  // @sic R5-213521 sic@
          // @siclog "Step 14a1" siclog@            INVITE
          v_InviteRequestWithSdp := f_IMS_MOCallSetup_NR5GC_A_4_1a_Step1(v_CalleeUri);        // @sic R5-225925, R5-236507: A_4_1 -> A_4_1a sic@
          f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 14a1");
          // @siclog "Step 14a2-14a15" siclog@      Steps 2 to 12 of A.4.1a
          f_IMS_MOCallSetup_NR5GC_A_4_1a_Step2_12(v_InviteRequestWithSdp);                    // @sic R5-225925, R5-236507: A_4_1 -> A_4_1a sic@
        } else {
          // @siclog "Step 14b1" siclog@            INVITE
          v_InviteRequestWithSdp := f_IMS_MOCallSetup_NR5GC_A_4_2a_Step1(v_CalleeUri);        // @sic R5-236507: A_4_2 -> A_4_2a sic@
          f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 14b1");
          // @siclog "Step 14b2-14b8" siclog@       Steps 2 to 8 of subclause A.4.2a
          f_IMS_MOCallSetup_NR5GC_A_4_2a_Step2_8(v_InviteRequestWithSdp);                     // @sic R5-236507: A_4_2 -> A_4_2a sic@
      }
      }
      [] t_Timer.timeout {
        f_IMS_CC_AbortCall();         // @sic R5s230086 sic@
      }
    }

    f_IMS_TestBody_Set(false);

    // Call release and postamble:
    f_IMS_CC_PostambleWithCallReleaseMT(IPCAN_MO_SpeechCall);
  }<<SKIPPED CODE>>



After Change:
	<<SKIPPED CODE>>
  function f_TC_7_1_NR5GC_IMS1() runs on IMS_PTC
  { // MTSI MO Voice Call / 503 Service Unavailable / 5GS
    var template (present) SipUrl v_CalleeUri := f_IMS_INVITE_CalleeUriRX();    // as used per default in the INVITE and in this test case for the ACK as the UE has not gotten any callee's contact URI
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var INVITE_Request v_InviteRequest;
    var template (value) MessageHeader v_MessageHeader503;
    var template (present) MessageHeader v_MessageHeaderACK;
    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;
    var integer v_RetryAfter := 20;   // @sic R5-210882: 60s -> 20s sic@
    var float v_RetryAfterMin := f_IMS_SetTimerToleranceMin(int2float(v_RetryAfter));
    var boolean v_InviteWithPreconditions;
    var boolean v_MOVoiceCallEstablished := true;    
    
    timer t_Timer;

    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED);
    
    f_IMS_TestBody_Set(true);

    // @siclog "Step 1" siclog@
    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall);

    // @siclog "Step 2 - 7" siclog@
    v_InviteWithPreconditions := fl_IMS_MOCallSetup_CheckInviteWithPreconditions();

    if (v_InviteWithPreconditions) {  // @sic R5-213521 sic@
      // @siclog "Step 8a" siclog@              INVITE
      v_InviteRequestWithSdp := f_IMS_MOCallSetup_NR5GC_A_4_1a_Step1(v_CalleeUri);               // @sic R5-225925, R5-236507: A_4_1 -> A_4_1a sic@
    } else {
      // @siclog "Step 8b" siclog@              INVITE
      v_InviteRequestWithSdp := f_IMS_MOCallSetup_NR5GC_A_4_2a_Step1(v_CalleeUri);               // @sic R5-236507: A_4_2 -> A_4_2a sic@
    }
    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    // @siclog "Step 9" siclog@                 503 Service Unavailable
    v_MessageHeader503 := f_IMS_CommonResponse_503_MessageHeaderTX(v_InviteRequest.msgHeader, cs_RetryAfter(v_RetryAfter));
    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_InviteRequestWithSdp.RoutingInfo);
    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine503, v_MessageHeader503)));
    
    // @siclog "Step 10" siclog@                ACK
    v_MessageHeaderACK := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine503);
    f_IMS_ReceiveRequest(car_IMS_Ack_Request(cr_ACK_Request(v_CalleeUri, v_MessageHeaderACK)));
    
    f_IMS_PTC_ImsInfo_DialogRelease();

    // @siclog "Step 11 - 13" siclog@           Check: Does the UE (not) transmit INVITE request message
    f_Delay(v_RetryAfterMin);
    
    t_Timer.start(30.0);
    alt {                     // @sic R5-216210 sic@
      [] IMS_Server.check(receive(car_IMS_Invite_Request)) {
        t_Timer.stop;
        if (v_InviteWithPreconditions) {  // @sic R5-213521 sic@
          // @siclog "Step 14a1" siclog@            INVITE
          v_InviteRequestWithSdp := f_IMS_MOCallSetup_NR5GC_A_4_1a_Step1(v_CalleeUri);        // @sic R5-225925, R5-236507: A_4_1 -> A_4_1a sic@
          f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 14a1");
          // @siclog "Step 14a2-14a15" siclog@      Steps 2 to 12 of A.4.1a
          f_IMS_MOCallSetup_NR5GC_A_4_1a_Step2_12(v_InviteRequestWithSdp);                    // @sic R5-225925, R5-236507: A_4_1 -> A_4_1a sic@
        } else {
          // @siclog "Step 14b1" siclog@            INVITE
          v_InviteRequestWithSdp := f_IMS_MOCallSetup_NR5GC_A_4_2a_Step1(v_CalleeUri);        // @sic R5-236507: A_4_2 -> A_4_2a sic@
          f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 14b1");
          // @siclog "Step 14b2-14b8" siclog@       Steps 2 to 8 of subclause A.4.2a
          f_IMS_MOCallSetup_NR5GC_A_4_2a_Step2_8(v_InviteRequestWithSdp);                     // @sic R5-236507: A_4_2 -> A_4_2a sic@
      }
      }
      [] t_Timer.timeout {
        f_IMS_CC_AbortCall(); // @sic R5s230086 sic@
        v_MOVoiceCallEstablished := false;    
      }
    }

    f_IMS_TestBody_Set(false);

    // Call release if established and postamble:  
    if (v_MOVoiceCallEstablished){
       f_IMS_CC_PostambleWithCallReleaseMT(IPCAN_MO_SpeechCall);   
    } else{
       f_IMS_CC_Postamble(IPCAN_MO_IMS_Signalling);      
    }  
  }
<<SKIPPED CODE>>



MCC160 Implementation
	  function f_NR5GC_IPCAN_StartProcedure(IMS_IPCAN_CO_ORD_PORT p_Port,
                                        NR_CellId_Type p_NR_CellId,
                                        IMS_TestProcedure_Type p_ProcedureType,
                                        template (omit) IMS_TestConfiguration_Type p_TestConfiguration,
                                        boolean p_ConnectionEstablished) runs on NR5GC_PTC return TriggerResult_Type
  {
    var TriggerResult_Type v_Result := NORMAL;

    select (p_ProcedureType) {
      case (IPCAN_InitialRegistration) {
        f_NR5GC_IPCAN_InitialRegistration(p_NR_CellId);
      }
      case (IPCAN_EmergencyCall_NormalService,                                  // emergency call, normal service, uses 112 as default emg number
            IPCAN_EmergencyCall_OptionalPDNRelease) {
        if (isvalue(p_TestConfiguration)) {  // @sic R5-225426 sic@
          if (match(IPCAN_SpeechAndEmergencyCall, p_TestConfiguration)) {
             f_NR5GC_IPCAN_EmergencyAddVoice(p_Port, p_NR_CellId);
          }
        } else {
          f_NR5GC_IPCAN_EmergencyCall(p_Port, p_NR_CellId, p_ConnectionEstablished);  // @sic R5s220531 sic@
        }
      }
      case (IPCAN_EmergencyCall_LimitedService) {                               // emergency call, limited service
        /* nothing to be done */
      }
      case (IPCAN_MO_SpeechCall) {                                              // MO speech call
        v_Result := f_NR5GC_IPCAN_MO_SpeechCall(p_Port, p_NR_CellId);
      }
      case (IPCAN_MT_SpeechCall) {                                              // MT speech call
        v_Result := f_NR5GC_IPCAN_MT_SpeechCall(p_Port, p_NR_CellId);
      }
      case (IPCAN_MO_VideoCall) {                                               // MO video call
        v_Result := f_NR5GC_IPCAN_MO_VideoCall(p_Port, p_NR_CellId);                        // @sic R5s210877 sic@
      }
      case (IPCAN_MT_VideoCall) {                                               // MT video call
        v_Result := f_NR5GC_IPCAN_MT_VideoCall(p_Port, p_NR_CellId);
      }
      case (IPCAN_MO_AddVideo, IPCAN_MT_AddVideo) {                             // add video to existing speech call
        v_Result := f_NR5GC_IPCAN_AddVideo(p_Port, p_NR_CellId);
      }
      case (IPCAN_ReleaseVideo) {                                               // Release video but keep speech call
        v_Result := f_NR5GC_IPCAN_ReleaseVideo(p_Port, p_NR_CellId);
      }
      case (IPCAN_MO_IMS_Signalling) {                                          // MO IMS signalling
        f_NR5GC_IPCAN_MO_IMS_Signalling(p_NR_CellId);
      }
      case (IPCAN_MT_IMS_Signalling) {                                          // MT IMS signalling
        f_NR5GC_IPCAN_MT_IMS_Signalling(p_Port, p_NR_CellId);
      }
      case (IPCAN_XCAP_Signalling) {                                            // XCAP signalling
        f_NR5GC_IPCAN_XCAP_Signalling(p_NR_CellId);
      }
      case (IPCAN_AdditionalIPAddress) { // @sic R5-213400 sic@
        f_NR5GC_IPCAN_AdditionalIPAddress();
      }
    }
    return v_Result;
  }

	  function f_IMS_CC_PostambleWithCallReleaseMT(IMS_TestProcedure_Type p_TestProcedure) runs on IMS_PTC
  {
    var template (omit) SipUrl v_ContactUrlUE := f_IMS_PTC_ImsInfo_GetContactUrlUE(dialog);

    if (ispresent(v_ContactUrlUE)) {
    if (f_IMS_PTC_ImsInfo_DialogIsEstablished()) {

      v_ContactUrlUE := f_IMS_PTC_ImsInfo_GetContactUrlUE(dialog);

      f_Delay(2.0);      // keep the call for some time

      f_IMS_CallReleaseMT(valueof(v_ContactUrlUE));
    }
    f_IMS_CC_Postamble(p_TestProcedure);
  }




